

## Mapping Roanoke Island Revisited: An OpenStreetMap (OSM) Solution

Barbara Okerson, Anthem Inc.


### ABSTRACT

In a previous presentation, SAS® was used to illustrate the difficulty and solutions for mapping small pieces of coastal land that are often removed from map boundary files to smooth boundaries. Roanoke Island, one of the first areas of the current United States to be mapped (1585), was used as an example since it is smoothed out of many current maps. While these examples isolated Roanoke Island, they didn't provide detail beyond city names on the map.

Originally limited to SAS Visual Analytics, SAS now makes available background maps with street and other detail information for SAS/GRAPH® using open source map data from OpenStreetMap (OSM). This paper will review the previous solutions and then look at how to map Roanoke Island using SAS/GRAPH and OSM.

### INTRODUCTION AND BACKGROUND


One of the first maps of the present United States was John White's 1585 map of the Albemarle Sound and Roanoke Island, North Carolina--my present home. Current maps often exclude areas such as Roanoke Island as the emphasis has changed from accurate boundaries for navigation to ease of data display. While traditional SAS maps often exclude Roanoke Island, both the SAS traditional and GfK map solutions can be used to isolate and map this area. Illustrations below show the early John White map (Roanoke Island prominent) and a current North Carolina tourism map (Roanoke Island missing).



**Figure 1. Roanoke Island maps.**

As maps began to be generated by computer, much of the detail provided by the cartographer was lost. Maps were output by line printers that were not capable of the resolution provided by earlier hand-created maps. While coordinates were stored as data and were accurate, these early computer maps were dependent upon memory size and graphics capabilities. These maps quickly became a base for displaying other information geographically. Because the ability to display data became more important than the preciseness of the boundaries, edges became smoothed and even the accuracy of the early computer maps was compromised. The advantage was in the display of multiple layers of information upon a single base; the disadvantage in the disappearance of small areas like Roanoke Island.

While SAS defaults also produce maps that contribute to the disappearance of Roanoke Island and other small land areas, fortunately SAS provides solutions that can include these in SAS output. The solution below shows that with the inclusion of the GfK map datasets, SAS provides accurate and consistent datasets from which to subset small areas such as coastal islands.



**Figure 2. Dare County and Roanoke Island with SAS GfK map datasets.**

Since SAS 9.3, digital, vector-based base maps supplied by GfK Geomarketing are available in SAS in addition to the traditional SAS map data sets. While both the SAS traditional map datasets and the GfK maps continue to be updated, the GfK maps are intended to eventually replace the traditional maps since they provide both: 1) a single source for map data throughout the world and 2) increased resolution for many boundaries.

While the maps above provide an accurate rendering of Dare County and Roanoke Island and a base platform for mapping data, GfK does not provide information for mapping streets, roads and other infrastructure. While users can license or purchase this information from many vendors, SAS provides and interface to the free OpenStreetMap data.

## OPENSTREETMAP (OSM)

OpenStreetMap is a source of mapping data built by a community of mappers that contribute and maintain data about roads, trails, cafés, railway stations, and much more, all over the world. OSM is open data and can freely be used for any purpose as long as OpenStreetMap and its contributors are credited. In other words, OSM is a collaborative project to create a free and editable map of the world. It is supported by the OpenStreetMap Foundation, a non-profit registered in England and Wales.

The OSM project was started because geographic data is not free in many parts of the world. For example, Google maps data is copyrighted by many organizations and licensed by Google, therefore not free to use without permission. Created by Steve Coast in the UK in 2004, OSM was inspired by the success of Wikipedia and the predominance of proprietary map data in the UK and elsewhere. OSM currently has over two million registered users who collect data using manual survey, GPS devices, aerial photography, voice recorders and other free sources.

## SAS OSM ANNOTATE GENERATOR

SAS provides a tool called the OSM Annotate Generator to enable SAS users to use OSM data in SAS. While the OSM Annotate Generator is written in SAS/AF® SCL code, it does not require a SAS/AF license to run. The OSM Annotate Generator creates an annotate data set containing a static background map that can be used with PROC GMAP. This tool retrieves the required background map images from the server and stores them on your local disk while creating the annotate data set that uses the images.

The area displayed on the map is specified either through: 1) a call to or subset of a map in SAS map data sets (either traditional or GfK), or 2) through inclusion of latitude/longitude information. The resulting background OSM maps can display data in addition to the OSM information. In both situations, PROC GMAP uses the data sets that were created.

As mentioned above, the OSM Annotate Generator is a SAS/AF SCL application. Values are passed to the Annotate Generator by setting special macro variables with a series of %lets. Here are the macro variables accepted by the application:

- **\_inmapds** – Input map data set (polygonal data)  
OR **\_inannods** – Input annotate data set.
- **\_imageloc** – Location where the map tile set images will be downloaded and saved.
- **\_outannoOSMds** – Output annotate data set containing the background map.
- **\_xpixels** – xpixel size of the background image. Defaults to 800.
- **\_ypixels** – ypixel size of the background image. Defaults to 600.
- **\_autoproject** – Turn on the projection to Mercator to match what is needed for OSM.
- **\_outmapds** – Output map data set
- **\_outannods** – Output annotate data set.
- **\_backgroundonly** – Turn on the background only option.
- **\_tileserv** – Location of the map tile server being used if other than the default server.
- **\_copyright** – Set the text of the copyright string for alternate **\_tileserv** settings.
- **\_proxy** – Set your proxy server if needed.

To run the OSM Annotate Generator, specify the following:

```
PROC DISPLAY cat=OAG.osmmapanno.osm_anno_gen.scl;  
Run;
```

Values set with the macro variables stay set until they are changed or cleared. If you forget to clear them, you may get unexpected results or errors. SAS provides a program called `OAG_reset.sas` that should be run before each program. The OSM Annotate generator and supporting programs are available at: <http://support.sas.com/rnd/papers/sasgf13/GooglelikeMaps.zip>.

## OSM WITH JAVA MAP APPLLET

The JAVA Map Applet has been modified in SAS 9.4 to support OpenStreetMap map data. The resulting maps can be panned and/or zoomed when the JAVA device is specified. As the map is panned or zoomed, additional detail comes into view. Static images can also be created with the JAVA device when `JAVAIMG` is specified. Syntax changes to Proc GMAP in SAS 9.4 are required for using the JAVA Map Applet. It is also possible to use it with earlier versions of SAS but some modifications to the code need to be made. This JAVA Map Applet can be downloaded from: <http://support.sas.com/rnd/datavisualization/mapsonline/html/tools.htm>.

The syntax below is used to enable this extended functionality, depending on which version of SAS you are using. With SAS 9.4, there are options that are added to the PROC GMAP CHORO statement that enable the OpenStreetMap capability. This syntax works with both the JAVA device and the JAVAIMG device.

```
CHORO variable / SHOWOSM <=>(<STYLE=osmstyle> <autoproject>)>;  
  
Where: osmstyle is SASMAPNIK or SASMAPNIK_LITE
```

If you specify the OSM option (SHOWOSM) without any suboptions, the GMAP procedure by default uses the SASMAPNIK style and does not project the map data. To use the JAVA device with OSM before SAS 9.4, you must enable the OSM capability with the 'parameters' option on the ODS HTML statement.

```
ODS HTML .... parameters=("OSMSTYLE"="style" <"AUTOPROJECT"="TFvalue"> );  
  
Where: style is SASMAPNIK or SASMAPNIK_LITE and TFvalue is TRUE or FALSE  
(Converts unprojected latitude/longitude to Mercator projection)
```

## MAPPING EXAMPLES WITH SAS OSM ANNOTATE GENERATOR

The examples that follow will look at the creation of maps for Dare County, North Carolina, especially Roanoke Island using the OSM Annotate Generator. Examples will include:

- Dare County area with streets and highways
- Roanoke Island streets and highways
- Roanoke Island terrain
- Subdivision map within Roanoke Island
- CHORO map with OSM

### EXAMPLE ONE – DARE COUNTY AND ENVIRONS

Dare County is the easternmost county in the state of North Carolina. The county is named after Virginia Dare, the first child born in the Americas to English parents, who was born in what is now Dare County. As of the 2010 census, the population of the county was 33,920; however the county's tourism industry results in a large seasonal population with an average daily population from June through August of between 225,000 to 300,000 people. The county seat is Manteo on Roanoke Island which is about 200 miles east of Raleigh.

The map below is created by subsetting the SAS GfK North Carolina map for Dare County and overlaying that information transparently over the OSM background. The GfK coordinates dictate the OSM area, however the background image will also include the nearby area.




Figure 3. Dare County area with OSM.

Here is the code used to create the map above. Note the use of the OSM Annotate Generator macro values. The pattern color A00000000 equates to completely transparent so all of the OSM features remain visible. This color pattern is part of the RGBA color codes available beginning with SAS 9.3. Color names are in the form of ARRGGBBAA, as below:

- A – RGBA color specification.
- RR – red component.
- GG – green component.
- BB – blue component.
- AA – transparency component.

```

/* File, Directory and Program Locations*/
filename odsout '/cma/nshare/bokerson/OSMMaps';
libname OAG '/cma/nshare/bokerson/OSMMaps/AFpgm';

/* reset the macro variables */
%include '/cma/nshare/bokerson/OSMMaps/AFpgm/OAG_reset.sas';

/* Subset map for Dare County*/
data daremap; set mapsgfk.us_counties (where=(statecode='NC' and county=055));
  x=long; y=lat;
run;

/* Set the macro variables for the OSM Generator */
%let _inmapds=daremap;
%let _outmapds=work.mdaremap;
%let _outannoOSMds=work.mannoosm;
%let _imageloc='/cma/nshare/bokerson/OSMMaps/';
%let _xpixels=2400;
%let _ypixels=1800;
%let _autoproject='Y';

goptions reset;
/* Set xpixels, ypixels to the size to be used */
goptions xpixels=&_xpixels ypixels=&_ypixels;

/* Generate background image */
PROC DISPLAY cat=OAG.osmmapanno.osm_anno_gen.scl; run;


/* Generate the map */
GOPTIONS DEVICE=png;
ODS LISTING CLOSE;
ODS HTML path=odsout body="ChoroDCGfK_png..html";
pattern; /*reset pattern statements */
pattern v=me c=A00000000 r=100;

PROC GMAP map=&_outmapds anno=&_outannoOSMds data=daremap all;
  id county;
  choro statecode /nolegend name="ChoroDCGfK";
run;
quit;
ODS HTML CLOSE;
ODS LISTING;

```

## EXAMPLE TWO – ROANOKE ISLAND

The map below is also created by subsetting the SAS GfK North Carolina map for Dare County and overlaying that information transparently over the OSM background. For this map, the map is further subset by segment to isolate Roanoke Island. Additionally, graphics size options are used to retain the resolution while subsetting. Note the increase in visible features such as the airport when subsetting to a smaller geographic area.



**Figure 4. Roanoke Island with OSM.**

Here is the code used to create the Roanoke Island map above. The `xpixels`, `ypixels`, `hsize` and `vsize` options were set to set resolution for the map and map size of the display. The `segment subset` is set on both the data and map data sets. Other options are the same as in the Dare County map above.

```

/* Set the macro variables for the OSM Generator */
%let _inmapds=daremap;
%let _outmapds=work.mdaremap;
%let _outannoOSMds=work.mannoosm;
%let _imageloc='/cma/nshare/bokerson/OSMMaps/';
%let _xpixels=4800;
%let _ypixels=3600;
%let _autoproject='Y';

goptions reset;
goptions xpixels=&_xpixels ypixels=&_ypixels hsize= 5in vsize= 5in;

/* Generate background image */
PROC DISPLAY cat=OAG.osmmapanno.osm_anno_gen.scl; run;

/* Generate the map */
GOPTIONS DEVICE=png;
ODS LISTING CLOSE;
ODS HTML path=odsout body="ChoroRIGfK_png..html";
pattern; /*reset pattern statements */
pattern v=me c=A000000000 r=100;
PROC GMAP map=mdaremap(where=(segment=4)) anno=mannoosm
data=daremap(where=(segment=4)) all;
  id county;
  choro statecode /nolegend name="ChoroRIGfK";
run;quit;
ODS HTML CLOSE;
ODS LISTING;

```

### EXAMPLE THREE – ROANOKE ISLAND TERRAIN

What OSM provides are background tile sets through a map tile server. More than one map tile server is available. Use of these will require using the correct copyright message on your map and that you have a legal right to use the tile set that you have selected. These will run slower in SAS because they are not already set up for SAS use. To use these, two macro variables not needed in the above maps will be used: `_tileserv` and `_copyright`. By default, `_tileserv` is set to the SAS provided tile server. To point it to an alternate server, specify the following:

```
%let _tileserv= <tileserver>;
```

When specifying a `_tileserv`, you must also specify the proper copyright message required by the owner of the map tile set. `_copyright` is used for this.

```
%let _copyright= <copyright-statement>;
```

The map below is uses the same subsetting of Roanoke Island as the previous map, but uses map tile set data from Stamen Design. Stamen is a San Francisco design and development studio focused on data visualization and map-making that provides three free tile sets that can be used with OpenStreetMap data worldwide: Toner, Terrain, and Watercolor. The example map images below use the Terrain map tile set for Roanoke Island.



**Figure 5. Roanoke Island terrain maps with OSM.**

The code below shows the macro values that are set to create the Roanoke Island map above. Roanoke Island has little variation in terrain; terrain maps are more useful for mapping larger geographic areas or areas with greater landscape variation. Other than the macro values and renaming the output files, the code is the same as in Figure 4. Note the call to the web location for the Stamen terrain tile set. Note also the background option on the `_tileserv` macro statement in the code. That produces the first image. No option for the macro variable would produce the image on the right with labels and lines. Options available for the terrain set are:

- Standard terrain (default)
- Labels
- Lines
- Background

Using any of the options limits the output to just that option. For example, the lines options produces an image without boundaries – just the streets and highways. Please again note that for the free use of these tile sets, the copyright statement is required.

```

/* Set the macro variables for the OSM Generator */
%let _inmapds=daremap;
%let _outmapds=work.mdaremap;
%let _outannoOSMds=work.mannoosm;
%let _imageloc='/cma/nshare/bokerson/OSMMaps/';
%let _xpixels=4800;
%let _ypixels=3600;
%let _autoproject='Y';
%let _tileserv=tile.stamen.com/terrain-background; /*terrain */
%let _copyright="Map tiles by Stamen Design, Data by OpenStreetMap.";

```

To create the second map, modify the `_tileserv` macro variable statement as follows:

```

%let _tileserv=tile.stamen.com/terrain;

```

#### EXAMPLE FOUR – ROANOKE ISLAND SUBDIVISION

The subdivision map below is created in SAS without using a SAS map data set as a base. Rather the map uses points and their latitude/longitude to determine the boundaries of the mapped space. Any points where this information is available can be provided to the OSM Annotate Generator to create a map. Here three locations in the Pirates Cove subdivision of Manteo on Roanoke Island were provided to create a residences data set to be used with the OSM tile set. Public records were used to identify the latitude and longitude of the three selected points. The size of the area selected determines the number of streets and features that are labeled. Here the main roads, bridge, and water bodies are labeled. Here the main roads, bridge, and water bodies are labeled.




Figure 6. Roanoke Island Subdivision with OSM



Here is the code to create the subdivision map. Note the lack of a SAS map file. Therefore the macro variable `_inmapds` is replaced with the `_inannods` macro variable. The OSM Annotate Generator requires either `_inmapds` or `_inannods` but cannot have both. In the residences data set, `x`= the longitude and `y` = the latitude.

```

/* Subdivision Residences */
data residences;
  input y x rec_no;
cards;
35.8997 -75.6403 1
35.8937 -75.6448 2
35.9096 -75.6463 3
;
run;

/* Create annotate data set of dots at locations */
%make_dots(annodots, residences, 'RED');

/* Set the macro variables for the OSM Generator */
%let _backgroundonly='YES';
%let _inannods=annodots;
%let _outannods=mannodots;
%let _outmapds=mmap;
%let _outannoOSMds=work.mannoosm;
%let _imageloc='/cma/nshare/bokerson/OSMMaps/';
%let _xpixels=800;
%let _ypixels=600;
%let _autoproject='Y';
%let _tileserv="SAS";

options reset;
options xpixels=&xpixels ypixels=&ypixels xmax=12in ymax=9in;

/* Generate background image */
PROC DISPLAY cat=OAG.osmmapanno.osm_anno_gen.scl; run;

/*Combine the annotate data sets - The background and the dots*/
data bothanno;
  set &_outannoOSMds mannodots;
run;

/* Generate the map*/
GOPTIONS DEVICE=png;
ODS LISTING CLOSE;
ODS HTML path=odsout body="PC_ALT_png..html";
pattern;
pattern v=me c=A00000000 r=100;
PROC GMAP map=&_outmapds anno=bothanno data=&_outmapds all;
  id id;
  choro id /nolegend name="PC_ALT";
run;
quit;
ODS HTML CLOSE;
ODS LISTING;

```

The code above includes a call to a macro program (`make_dots`) that creates the dots that are included on the map. Three parameters need to be provided: the name of the output data set, the data set being fed into the macro (contains latitude/longitude locational information for the dots), and the color for the dots. The macro expects to get latitude and longitude information as `y` for latitude and `x` for longitude. The code for this macro is provided below. This code can be modified to add additional parameters and annotate variables. Here is the code used for the above map.

```

%macro make_dots(annodata, inputdata, color);
data &annodata;
  length function color $ 8 style $20;
  retain xsys ysys '2' hsys '3' when 'a' size .7;
  set &inputdata;

/* All dots are filled 360-degree pies. */
function='pie';
rotate=360;
style='psolid';
color=&color;
output;

/* Use the following to outline each dot */
color="gray55"; style='pempty';
output;
run;
%mend;

```

### EXAMPLE FIVE – CHOROPLETH MAP WITH OSM ANNOTATE GENERATOR

One of the most useful applications of OSM maps and the OSM Annotate Generator is the ability to map data on a base map that includes highways and other features without using specialized GIS software. To create a choropleth map, a response data needs to be created and/or processed. This data set needs to be tied to a SAS/GRAPH map data set. Note also the addition of a legend on the map. Legends follow general SAS map legend guidelines and can be positioned inside or outside the image.




Figure 7. Choropleth Map with OSM.

Partial code is listed below, illustrating the patterns used, the legend statement and the GMAP procedure. It is important to use transparent colors when creating choropleth maps with the OSM map tiles so as not to lose the OSM information. Chorseg is the variable that identifies the choropleth areas and was created for illustrative purposes.

```

/* Set transparent colors */
pattern1 v=solid repeat=1 c=A00ffffaa;
pattern2 v=solid repeat=1 c=Afe0000aa;

/* Create a legend */
legend1 label=(position=top justify=center
  font="albany amt" height=1 " ")
  shape=bar(.1in,.1in)
  value=(justify=left height=1 font="albany amt"
    'Outlying Dare County' 'Roanoke Island');

/* Draw the map */
PROC GMAP map=&_outmapds anno=&_outannoOSMds data=mymap all;
  id chorseg county;
  choro chorseg /legend=legend1 name="ChoroDCGfK";
run; quit;

```

## CONCLUSION

The availability of OpenStreetMap functionality in SAS through the OSM Annotate Generator provides tools to add streets and other landmarks to SAS/GRAPH PROC GMAP programs with only simple modifications. By modifying the resolution and map area, the amount of detail on the map can be controlled. Additional tile sets that work with OpenStreetMap are available for use that allow terrain and elevation mapping. For an area like Roanoke Island that is removed from standard SAS maps, the ability to not only map the island, but map what is on the island is greatly appreciated.

## REFERENCES

- "About Dare County." Dare County. Archived from the original on 3 April 2012. <https://web.archive.org/web/20120403130938/http://www.darenc.com/general/mainabout.asp>. Accessed Sept 26, 2017.
- Coast, S. 2006. "Yahoo! aerial imagery in OSM". *OpenGeoData*. Available at: <https://blog.openstreetmap.org/2006/12/04/yahoo-aerial-imagery-in-osm/>
- "Learn OSM: Learn OpenStreetMap Step by Step." Available at: <http://learnosm.org/en/>. Accessed Sept 27, 2017.
- Massengill, D. 2013. "'Google-Like' Maps in SAS®." SAS Global Forum 2013 Proceedings. Available at: <https://support.sas.com/resources/papers/proceedings13/377-2013.pdf>.
- Native Heritage Project. "John White Map Chowan Fort Discovery – Analysis," <https://nativeheritageproject.com/2012/05/07/john-white-map-chowan-fort-discovery-analysis/>. Accessed Sept 2, 2016.
- North Carolina Map. [http://www.hotelfun4kids.com/hotelandresort/ushotels/north\\_carolina.htm](http://www.hotelfun4kids.com/hotelandresort/ushotels/north_carolina.htm). Accessed June 27, 2016.
- Okerson, B. 2016. "Mapping Roanoke Island: from 1585 to present." SAS Global Forum 2017 Proceedings. Available at: <http://support.sas.com/resources/papers/proceedings17/0895-2017.pdf>.
- "OpenStreetMap." <https://www.openstreetmap.org/about>. Accessed Sept 11, 2017.
- SAS Graph 9.4: Mapping Reference. SAS Institute Inc. Available at: <http://support.sas.com/documentation/cdl/en/grmapref/69722/PDF/default/grmapref.pdf>.
- "Stamen." <http://wiki.openstreetmap.org/wiki/Stamen>. Accessed Sept 26, 2017.

## ACKNOWLEDGMENTS

A special thanks goes to Darrell Massengill and Jeff Phillips of SAS for their assistance in getting started with the SAS OSM solution.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Barbara B. Okerson, Ph.D., CPHQ, FAHM  
Health Information Manager  
Health Care Analytics  
Anthem, Inc.  
barbara.okerson@anthem.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.